

UR1986_57

1986/57. SEDSAM: a FORTRAN program for size of sediment analysis
(Revision 2)

J. W. Hudspeth

Abstract

SEDSAM is a program which analyses the results of size separation of the particles of a sediment. Mean, sorting, skewness and kurtosis are calculated (if the data permits), the formulae employed being controlled by the highest and lowest cumulative percentage weights. Linear interpolation is used.

INTRODUCTION

The program SEDSAM was written in response to a need for sediment grain size distribution analysis. The program calculates various parameters directly from the weights on the sieves without recourse to graph plotting. The user may elect to include or exclude the weight fraction which has passed the finest sieve. Alternatively, the user may input the cumulative percentage weights passing or retained for each sieve size.

Various formulae have been employed for calculation of the required parameters and the program automatically chooses the most efficient (King, 1966, p.283) formulae which may be used. The program is written in FORTRAN and runs interactively through a .CSS on the Department of Mines Perkin Elmer minicomputer. The program has been revised in response to changing and differing user requirements. The current revision allows users to input data as cumulative weight percentages passing or retained, as alternatives to weight retained on the sieve. The phi or mm option for sieve size remains, as does the option to include or exclude that fraction passing all sieves. Users may now elect to have the cumulative percentage weight retained data stored in a file so that sediment analysis curves may be easily plotted using a plotting program.

RUNNING THE PROGRAM

The program is run interactively and the user initiates the run by typing 'SEDSAM' to call a .CSS of that name. The user is asked if the results are to be plotted. Respond in the affirmative if you wish the cumulative weight percentages and corresponding sieve sizes to be written to a file for later plotting or other use, and you will then be asked to specify a plot file name up to eight characters long and not including an extension. The extension .SSP is added automatically by the program, and for the duration of the interactive session results will be written to a file of the specified name (provided a file of that name with extension .SSP does not already exist on your account).

Users are then asked for sample identification and are allowed twenty characters but users who have answered in the affirmative to the plot question should restrict themselves to less than eleven characters, as the remainder of the characters will be used by the program for internal housekeeping. In response to the next prompt, the user should enter the integer value equal to the number of sieves used. The program uses phi values for calculation and output but users have the option to input sieve sizes in millimetres by giving the appropriate response to the next prompt. The user is then repeatedly prompted for the sieve size (as phi or millimetre value, depending on the choice previously made by the user) and

the corresponding retained weight (or alternative, if specified). Sieve sizes must be entered in correct order from coarse to fine. If this condition is not met the program will write a message to that effect and return the user to data pairs entry starting point. After the number of pairs of values entered is equal to the number of sieves specified, the program will request the weight passing all sieves if input is by retained weight, otherwise the question is not asked as the program already has the information to calculate the fraction passing all sieves. The weight passing all sieves will then be displayed on the screen as a percentage of the total sediment weight and the user will choose whether to include this fraction or exclude it and readjust the percentage weights held on the sieves to total one hundred percent, or run the analysis twice - once each way. The program then calculates the mean and, if it has sufficient information, also calculates sorting, skewness and kurtosis. Intermediate phi values are obtained by linear interpolation.

In the case that more than 16% sediment by weight is passed by all sieves (unless this is to be ignored) or is left on the coarsest sieve, then no calculations are carried out and an appropriate message appears on the screen. After the program has completed its task with a sample, the user will be prompted for another sample by "AGAIN? (N/Y)". Responding with "N" will terminate the program but responding with "Y" will take the user to the start of the program for another sample. The sample identification, phi values and corresponding cumulative weight percentages, mean, skewness, sorting and kurtosis and certain messages will be printed (if appropriate).

REFERENCE

KING, C. A. M. 1966. *Techniques in geomorphology*. Edward Arnold Ltd : London.

[5 September 1986]

APPENDIX 1

SEDSAM program listing (revision 2)

```

C* * * * *
COMMON /DATA/ PHI(30),PCENT(30),N
CHARACTER*1 ANS
CHARACTER*3 SIZE
CHARACTER*4 EXT
CHARACTER*8 FNAME
CHARACTER*12 PFNAME
CHARACTER*20 SAMPLE
CHARACTER*30 INPUT
REAL CUMWT(30),MEAN,KURT
LOGICAL V95,V97,ADJUST,XMM,BOTH,WEIGHT,CPRET,CPASS,PLOT,MPLOT
MPLOT=.FALSE.
WRITE(5,4)
4 FORMAT(1X,'DO YOU WISH TO PLOT ANY RESULTS THIS SESSION?')
6 READ(5,55) ANS
IF(ANS.EQ.'Y') THEN
PLOT=.TRUE.
ELSE
PLOT=.FALSE.
IF(ANS.NE.'N') THEN
WRITE(5,808)
GO TO 6
ENDIF
ENDIF
10 IF(PLOT.AND..NOT.MPLOT) THEN
13 WRITE(5,14)
14 FORMAT(1X,'ENTER PLOT FILE NAME WITHOUT EXTENSION...8 CHARS MAX')
READ(5,16) FNAME
16 FORMAT(A8)
JINDEX=INDEX(FNAME,' ')
IF(JINDEX.EQ.1) GO TO 13
IF(JINDEX.EQ.0) THEN
PFNAME=FNAME//'.SSP'
ELSE
PFNAME=FNAME(1:JINDEX-1)//'.SSP'
ENDIF
OPEN(UNIT=7,FILE=PFNAME,TYPE='INDEX',RECL=20,STATUS='NEW')
ENDIF
SIZE='PHI'
V95=.FALSE.
V97=.FALSE.
ADJUST=.FALSE.
XMM=.FALSE.
BOTH=.FALSE.
WEIGHT=.FALSE.
CPRET=.FALSE.
CPASS=.FALSE.
WRITE(5,40)
40 FORMAT(1X,'ENTER SAMPLE IDENTIFICATION (20 CHARS. MAX)')
READ(5,42) SAMPLE
42 FORMAT(A20)
WRITE(6,44) SAMPLE
44 FORMAT(////,1X,A20,/)

```

```

WRITE(7,48) SAMPLE
48 FORMAT(A20)
WRITE(5,50)
50 FORMAT(1X,'ENTER NUMBER OF SIEVES USED')
   READ(5,*) N
   WRITE(5,53)
53 FORMAT(1X,'SIEVE SIZE IN PHI OR MM?          (P/M)')
54 READ(5,55) ANS
55 FORMAT(A1)
   IF(ANS.NE.'M'.AND.ANS.NE.'P') THEN
   WRITE(5,56)
56 FORMAT(1X,'ANSWER P OR M')
   GO TO 54
   ENDIF
   IF(ANS.EQ.'M') THEN
   XMM=.TRUE.
   SIZE=' MM'
   ENDIF
   WRITE(5,62)
62 FORMAT(' STATE WHETHER INPUT IS WEIGHT, CUMULATIVE PERCENTAGE',
1/, ' RETAINED, CUMULATIVE PERCENTAGE PASSED.....W,R,P')
63 READ(5,55) ANS
   IF(ANS.NE.'W'.AND.ANS.NE.'R'.AND.ANS.NE.'P') THEN
   WRITE(5,67)
67 FORMAT(' ANSWER W OR R OR P')
   GO TO 63
   ENDIF
   IF(ANS.EQ.'W') THEN
   WEIGHT=.TRUE.
   INPUT=' RETAINED WEIGHT VALUES, '
   ENDIF
   IF(ANS.EQ.'R') THEN
   CPRET=.TRUE.
   INPUT=' CUM PERCENT RETAINED VALUES, '
   ENDIF
   IF(ANS.EQ.'P') THEN
   CPASS=.TRUE.
   INPUT=' CUM PERCENT PASSED VALUES, '
   ENDIF
89 WRITE(5,90) INPUT
90 FORMAT(1X,'ENTER SIZE AND CORRESPONDING',A30,/, ' ONE ',
1'PAIR OF VALUES AT A TIME AS PROMPTED.',/,1X,'SIEVE ',
2'SIZES MUST BE IN ORDER FROM COARSE TO FINE.')
   DO 100 J=1,N
   WRITE(5,95) SIZE, INPUT
95 FORMAT(1X,'ENTER ',A3,' VALUE AND',A30,
1,/, ' SEPARATE VALUES BY A COMMA')
   READ(5,*) PHI(J),WT
   IF(XMM) THEN
   PHI(J)=-ALOG10(PHI(J))/ALOG10(2.0)
   ENDIF
   IF((PHI(J).LT.PHI(J-1)).AND.(J.GT.1)) THEN
   WRITE(5,97)
97 FORMAT(1X,'PHI ORDERING ERROR - ORDER MUST BE COARSE TO FINE',
1/,1X,'START AGAIN')
   GO TO 89
   ENDIF
   IF(WEIGHT) THEN

```

```

IF(J.EQ.1) THEN
CUMWT(J)=WT
ELSE
CUMWT(J)=CUMWT(J-1)+WT
ENDIF
ELSE
IF(CPASS) PCENT(J)=100.0-WT
IF(CPRET) PCENT(J)=WT
ENDIF
100 CONTINUE
IF(WEIGHT) THEN
WRITE(5,105)
105 FORMAT(1X,'ENTER WEIGHT PASSED THROUGH ALL SIEVES')
READ(5,*) FINES
TOTWT=CUMWT(N)+FINES
FPCENT=100.0*FINES/TOTWT
DO 108 J=1,N
PCENT(J)=100.0*CUMWT(J)/TOTWT
108 CONTINUE
ELSE
FPCENT=100.0-PCENT(N)
ENDIF
WRITE(5,110) FPCENT
110 FORMAT(1X,'FRACTION PASSING ALL SIEVES IS ',F5.1,' PERCENT',
1/,10X,'DO YOU WISH TO IGNORE THIS FRACTION? (N/Y)',
2/,' . . . OR TRY IT BOTH WAYS? (B)')
111 READ(5,55) ANS
IF(ANS.NE.'N'.AND.ANS.NE.'Y'.AND.ANS.NE.'B') THEN
WRITE(5,114)
114 FORMAT(1X,'ANSWER N OR Y OR B')
GO TO 111
ENDIF
118 IF(ANS.EQ.'Y'.OR.BOTH) THEN
FACTOR=100.0/PCENT(N)
ADJUST=.TRUE.
ELSE
FACTOR=1.0
ENDIF
IF (ANS.EQ.'B') BOTH=.TRUE.
DO 130 J=1,N
PCENT(J)=FACTOR*PCENT(J)
WRITE(5,128) PHI(J),PCENT(J)
WRITE(6,128) PHI(J),PCENT(J)
128 FORMAT(1X,'PHI VALUE',F7.2,',', PERCENT',F6.1)
WRITE(7,129) PHI(J),PCENT(J)
129 FORMAT(2F10.3)
130 CONTINUE
IF(PCENT(1).GT.16.0) THEN
WRITE(5,135)
WRITE(6,135)
135 FORMAT(1X,'PERCENTAGE ON COARSEST SIEVE GREATER THAN 16 PERCENT'
1,/,', SUGGEST YOU RESIEVE THIS PORTION WITH COARSER SIEVES')
GO TO 800
ENDIF
IF(.NOT.ADJUST) THEN
IF(FPCENT.GT.16.0) THEN
WRITE(5,138)
WRITE(6,138)

```

```

138 FORMAT(1X,'PERCENTAGE PASSING ALL SIEVES GREATER THAN 16%'
1/, ' SUGGEST YOU RESIEVE THIS PORTION WITH FINER SIEVES OR',
2' IGNORE')
  IF(BOTH) THEN
  GO TO 118
  ELSE
  GO TO 800
  ENDIF
ENDIF
ENDIF
IF(PCENT(1).LT.5.0.AND.(ADJUST.OR.FPCENT.LT.5.0)) THEN
IF(PCENT(1).LT.3.0.AND.(ADJUST.OR.FPCENT.LT.3.0)) THEN
V97=.TRUE.
ENDIF
V95=.TRUE.
ENDIF
CALL LININT(84.0,PHI84)
CALL LININT(80.0,PHI80)
CALL LININT(75.0,PHI75)
CALL LININT(70.0,PHI70)
CALL LININT(65.0,PHI65)
CALL LININT(55.0,PHI55)
CALL LININT(50.0,PHI50)
CALL LININT(45.0,PHI45)
CALL LININT(35.0,PHI35)
CALL LININT(30.0,PHI30)
CALL LININT(25.0,PHI25)
CALL LININT(20.0,PHI20)
CALL LININT(16.0,PHI16)
CALL LININT(10.0,PHI10)
IF(V95) THEN
CALL LININT(5.0,PHI5)
CALL LININT(15.0,PHI15)
CALL LININT(85.0,PHI85)
CALL LININT(95.0,PHI95)
ENDIF
IF(V97) THEN
CALL LININT(3.0,PHI3)
CALL LININT(90.0,PHI90)
CALL LININT(97.0,PHI97)
ENDIF
IF(V95) THEN
MEAN=(PHI5+PHI15+PHI25+PHI35+PHI45
1+PHI55+PHI65+PHI75+PHI85+PHI95)/10.0
SKEW=(PHI16+PHI84-2*PHI50)/(2*(PHI84-PHI16))
1+(PHI5+PHI95-2*PHI50)/(2*(PHI95-PHI5))
KURT=(PHI95-PHI5)/(2.44*(PHI75-PHI25))
IF(V97) THEN
SORT=(PHI70+PHI80+PHI90+PHI97-PHI3-PHI10-PHI20-PHI30)/9.1
ELSE
SORT=(PHI84-PHI16)/4.0+(PHI95-PHI5)/6.6
ENDIF
WRITE(5,170) MEAN,SORT,SKEW,KURT
WRITE(6,170) MEAN,SORT,SKEW,KURT
170 FORMAT(1X,'      MEAN IS',F8.3,8X,' SORTING IS',F8.3,
1/, ' SKEWNESS IS',F8.3,8X,' KURTOSIS IS',F8.3,/)
ELSE
MEAN=(PHI16+PHI84)/2.0

```

```

WRITE(5,180) MEAN
WRITE(6,180) MEAN
180 FORMAT(1X,'MEAN IS ',F6.2,' NO SORTING,SKEWNESS OR KURTOSIS',
1' CALCULATED',/)
ENDIF
IF(BOTH.AND..NOT.AJUST) GO TO 118
800 WRITE(5,802)
802 FORMAT(1X,'AGAIN? (N/Y)')
804 READ(5,55) ANS
IF(ANS.EQ.'Y') THEN
IF(PLOT) MPlot=.TRUE.
GO TO 10
ENDIF
IF(ANS.EQ.'N') GO TO 910
WRITE(5,808)
808 FORMAT(1X,'ANSWER N OR Y')
GO TO 804
910 WRITE(6,911)
911 FORMAT(/)
IF(PLOT) THEN
CLOSE(UNIT=7,STATUS='KEEP')
ENDIF
STOP
END
SUBROUTINE LININT(XCENT,YPHI)
COMMON /DATA/ Y(30),X(30),N
DO 100 I=1,N
IF(XCENT.LT.X(I)) THEN
GO TO 200
ENDIF
100 CONTINUE
WRITE(5,105)
105 FORMAT(1X,'LININT ERROR')
GO TO 900
200 YPHI=Y(I-1)+(Y(I)-Y(I-1))*(XCENT-X(I-1))/(X(I)-X(I-1))
RETURN
900 WRITE(5,905) XCENT,X(I)
905 FORMAT(1X,'XCENT IS ',F6.2,' X(I) IS ',F6.2)
STOP
END

```